

Graphical Models

LUIS DAVID GARCÍA-PUENTE, SONJA PETROVIĆ, AND SETH SULLIVANT

ABSTRACT. The *Macaulay2* package *GraphicalModels* contains algorithms for the algebraic study of graphical models associated to undirected, directed and mixed graphs, and associated collections of conditional independence statements. Among the algorithms implemented are procedures for computing the vanishing ideal of graphical models, for generating conditional independence ideals of families of independence statements associated to graphs, and for checking for identifiable parameters in Gaussian mixed graph models. These procedures can be used to study fundamental problems about graphical models.

GRAPHICAL MODELS. A graphical model is a statistical model associated to a graph, where the nodes of the graph represent random variables and the edges of the graph encode relationships between the random variables. Graphical models are an important class of statistical models used in many applications (see the standard textbooks [L, W]) because of their ability to model complex interactions between several random variables, by specifying interactions using only local information about connectivity between the vertices in a graph.

There are two natural ways to specify a graphical model, through either conditional independence statements specified by the graph or via a parametric representation (often called a “factorization”). Every distribution that factors according to the graph satisfies the conditional independence statements implied by the graph. This leads to the question: Which distributions satisfy the conditional independence statements implied by the graph, but do not factor?

Once we specify the types of random variables under consideration (e.g. discrete random variables or Gaussian random variables) it is possible to address the questions in the preceding paragraph using (computational) algebraic geometry. Indeed, in these cases, the set of all probability distributions satisfying a family of conditional independence constraints is a semialgebraic set. For discrete random variables, that semialgebraic set is a subset of the probability simplex, and can be represented by a certain homogeneous ideal generated by quadrics. For Gaussian random variables, this set of distributions corresponds to a semialgebraic subset of the cone of positive definite matrices. Similarly, the parametrized family of probability distributions also is a semialgebraic set (of the probability simplex for discrete random variables, and of the cone of positive definite matrices for Gaussian random variables). This algebraic perspective has been studied by different authors [GSS, GMS, S], and the book [DSS] provides details.

The *Macaulay2* package *GraphicalModels* allows the user to compute the ideals of conditional independence statements for any collection of statements for discrete or Gaussian random variables. It can also compute the vanishing ideal of a graphical model in these cases. A number of auxiliary functions are useful for doing further analyses of graphical models.

For example, consider the directed acyclic graph G with five vertices $\{a, b, c, d, e\}$ and edge set $\{a \rightarrow d, b \rightarrow d, c \rightarrow d, c \rightarrow e, d \rightarrow e\}$. The following commands compute the associated conditional independence ideal for the set of global Markov statements, $CI_{\text{global}(G)}$, and the vanishing ideal I_G of the Gaussian graphical model on G .

```
Macaulay2, version 1.6
with packages: ConwayPolynomials, Elimination, IntegralClosure, LLLBases,
               PrimaryDecomposition, ReesAlgebra, TangentCone
i1 : needsPackage "GraphicalModels";
i2 : G = digraph{{a,d},{b,d},{c,{d,e}},{d,e}};
i3 : R = gaussianRing G;
i4 : gens R
o4 = {sa,a, sa,b, sa,c, sa,d, sa,e, sb,b, sb,c, sb,d, sb,e, sc,c, sc,d, sc,e,
-----
      sd,d, sd,e, se,e}
o4 : List
i5 : I = conditionalIndependenceIdeal(R,globalMarkov(G));
o5 : Ideal of R
i6 : J = gaussianVanishingIdeal(R);
o6 : Ideal of R
i7 : flatten degrees J
o7 = {1, 1, 1, 2, 3, 3}
```

GraphicalModels uses the package *Graphs* and a number of fundamental constructs and relationships associated with graphs. First we create a polynomial ring that contains the entries of the covariance matrix Σ of a jointly normal random vector as its indeterminates. Information about the underlying graph is stored in the polynomial ring. Hence some methods take just a ring as input, but require that it be created with `gaussianRing` or `markovRing` in the discrete case.

For directed acyclic graphs, it is known that

$$V(CI_{\text{global}(G)}) \cap PD_m = V(I_G) \cap PD_m$$

where $PD_m \subset \mathbb{R}^{\binom{m+1}{2}}$ is the cone of $m \times m$ positive definite symmetric matrices. In particular, the set of such matrices satisfying the conditional independence constraints equals the set of covariance matrices in the image of the parametrization. Unfortunately, this does not imply that $CI_{\text{global}(G)} = I_G$. In the case of Gaussian random variables, a larger ideal, the trek ideal T_G , generated by all subdeterminants of the covariance matrix that vanish on the model, and satisfying $CI_{\text{global}(G)} \subseteq T_G \subseteq I_G$ is sometimes equal to I_G (see [STD]), as the following example shows.

```
i8 : isSubset(I,J), I == J, J == trekIdeal(R,G)
o8 = (true, false, true)
o8 : Sequence
```

Similar computations can also be performed for graphical models with discrete random variables, and with other graph families. The mathematical explanation of these graphical models and their associated ideals appear in the remaining sections.

COMPUTING CONDITIONAL INDEPENDENCE IDEALS. Conditional independence constraints on discrete or Gaussian random variables translate to rank conditions on certain matrices associated to the probability densities. We briefly explain these constructions here and how to generate these constraints in *Macaulay2* using *GraphicalModels*; see [DSS, Ch. 3] for more detail.

Let $X = (X_1, \dots, X_n)$ be a discrete random vector where each random variable X_i has state space $[d_i] = \{1, 2, \dots, d_i\}$. Let $d = (d_1, \dots, d_n)$. A probability distribution for X is a tensor in $\mathbb{R}^{d_1} \otimes \dots \otimes \mathbb{R}^{d_n}$, all of whose coordinates are nonnegative and sum to one. The set of all such distributions is the probability simplex Δ_d . Let $p_{i_1 \dots i_n} = \mathbb{P}(X_1 = i_1, \dots, X_n = i_n)$ denote the probability of a primitive event. The polynomial ring in these quantities is created using the command `markovRing`.

```
i9 : d = (2,3,2); R = markovRing d;
i11 : gens R
o11 = {p1,1,1, p1,1,2, p1,2,1, p1,2,2, p1,3,1, p1,3,2, p2,1,1, p2,1,2, p2,2,1,
-----
      p2,2,2, p2,3,1, p2,3,2}
```

For $A \subseteq [n]$, let $X_A = (X_a)_{a \in A}$ be the subvector indexed by A . Let A, B, C be disjoint subsets of $[n]$. The conditional independence statement $X_A \perp\!\!\!\perp X_B \mid X_C$ holds if and only if the conditional distribution satisfies $\mathbb{P}(X_A = i_A, X_B = i_B \mid X_C = i_C) = \mathbb{P}(X_A = i_A \mid X_C = i_C) \cdot \mathbb{P}(X_B = i_B \mid X_C = i_C)$ for all i_A, i_B, i_C . This translates into vanishing 2-minors of certain matrices in the probabilities $p_{i_1 \dots i_n}$. Those matrices are computed with the function `markovMatrices`, and the ideal generated by the 2-minors is computed with `conditionalIndependenceIdeal`. In the following example, the two conditional independence statements are $X_1 \perp\!\!\!\perp X_2 \mid X_3$ and $X_1 \perp\!\!\!\perp X_3 := X_1 \perp\!\!\!\perp X_3 \mid X_\emptyset$. The ideal of vanishing minors has 7 quadratic generators.

```
i12 : S = {{{1},{2},{3}}, {{1},{3},{}}}; compactMatrixForm=false; markovMatrices(R,S)
o14 = { | p1,1,1 p1,2,1 p1,3,1 | | p1,1,2 p1,2,2 p1,3,2 | ,
      | p2,1,1 p2,2,1 p2,3,1 | | p2,1,2 p2,2,2 p2,3,2 |
      -----
      | p1,1,1 + p1,2,1 + p1,3,1 p1,1,2 + p1,2,2 + p1,3,2 | }
      | p2,1,1 + p2,2,1 + p2,3,1 p2,1,2 + p2,2,2 + p2,3,2 |
o14 : List
i15 : I = conditionalIndependenceIdeal(R,S); flatten degrees I
o15 : Ideal of R
o16 = {2, 2, 2, 2, 2, 2, 2}
```

In the statistics literature, there are three main lists of conditional independence statements associated to a graph G whose nodes correspond to random variables. For example, the list of *local Markov statements* of an undirected graph G is the set of conditional independence statements of the form $X_i \perp\!\!\!\perp X_{V \setminus \{i \cup N(i)\}} \mid X_{N(i)}$, where $N(i)$ is the set of neighbors of i in the graph G . The

methods `pairMarkov`, `localMarkov`, and `globalMarkov` compute the pairwise, local, and global Markov statements, respectively, for both directed and undirected graphs.

```
i17 : G = graph{{1,2},{2,3},{3,4},{4,5},{1,5}}; netList pack (3,localMarkov G)
+-----+-----+-----+-----+
o18 = |{{1}, {3, 4}, {5, 2}}|{{2, 3}, {5}, {4, 1}}|{{1, 2}, {4}, {5, 3}}|
+-----+-----+-----+-----+
      |{{1, 5}, {3}, {4, 2}}|{{2}, {4, 5}, {1, 3}}|
+-----+-----+-----+-----+
```

The first conditional independence statement produced is $X_1 \perp\!\!\!\perp (X_3, X_4) \mid (X_2, X_5)$. In the context of conditional independence, the graphical model consists of all distributions satisfying one of these collections of independence statements associated to the graph G .

A Gaussian random vector, $X = (X_1, \dots, X_n) \sim \mathcal{N}(\mu, \Sigma)$, is an n -dimensional random vector with state space \mathbb{R}^n and density function $f(x) = \frac{1}{(2\pi)^{n/2}(\det \Sigma)^{1/2}} \exp(-\frac{1}{2}(x - \mu)^T \Sigma^{-1}(x - \mu))$, where $\mu \in \mathbb{R}^n$ and $\Sigma = (\sigma_{s,t}) \in PD_n$, the cone of $n \times n$ symmetric positive definite matrices. The Gaussian random vector X satisfies the conditional independence statement $X_A \perp\!\!\!\perp X_B \mid X_C$ if and only if the submatrix $\Sigma_{AUC, BUC} := (\sigma_{s,t})_{s \in AUC, t \in BUC}$ has rank at most $\#C$. Hence, the set of all Gaussian random vectors satisfying a given collection of conditional independence statements yields a subset of PD_n that can be studied via a determinantal conditional independence ideal in the polynomial ring $\mathbb{Q}[\sigma_{s,t} : s, t \in [n]]$. This ring is generated using the command `gaussianRing`. Computations involving conditional independence ideals with Gaussian random variables were exemplified in the first section.

COMPUTING THE VANISHING IDEAL OF A MODEL. The fact that graphical models can be described in two possible ways (either by a recursive factorization of probability distributions or by conditional independence statements) corresponds to the algebraic principle that varieties can be presented either parametrically or implicitly. The *vanishing ideal* of a model is the set of homogeneous polynomial relations in the joint probability distributions (for discrete random variables) or in the variance-covariance parameters (for Gaussian random variables). *GraphicalModels* has the capability of computing the vanishing ideals of graphical models on directed graphs (for discrete random variables) and also of graphical models on directed, undirected, or mixed graphs (for Gaussian random variables). The vanishing ideal of an undirected graphical model for discrete random variables is a toric ideal and should be computed using the *Macaulay2* package *FourTiTwo*.

The method `discreteVanishingIdeal` implements this capability for graphical models on discrete random variables. For a directed acyclic graph G on discrete random variables, the graphical model consists of all distributions satisfying the recursive factorization property

$$P(X = i) = \prod_v P(X_v = i_v \mid X_{\text{pa}(v)} = i_{\text{pa}(v)}),$$

where the product runs over all vertices v of G and $\text{pa}(v)$ is the set of parents of v . Our implementation of this method does not compute the kernel of the corresponding ring map. Instead, the vanishing ideal is computed recursively using the factorization

$$P(X = i) = P(X_1 = i_1, \dots, X_{n-1} = i_{n-1}) \cdot P(X_n = i_n \mid X_{\text{pa}(n)} = i_{\text{pa}(n)}),$$

where $1, \dots, n$ is a topological ordering of the vertices of the directed acyclic graph G .

The following example computes the vanishing ideal of the graphical model $1 \rightarrow 2 \rightarrow 3 \rightarrow 4$ on four binary random variables. The vanishing ideal is minimally generated by 20 quadratic binomials.

```

i19 : G = digraph {{1,{2}}, {2,{3}},{3,{4}},{4,{}}}; R = markovRing (2,2,2,2);
i21 : I = discreteVanishingIdeal (R,G); betti mingens I
o21 : Ideal of R
      0 1
o22 = total: 1 20
      0: 1 .
      1: . 20
o22 : BettiTally

```

According to [GSS], the vanishing ideal of a graphical model on discrete random variables is the *distinguished component* of the conditional independence ideal described by the Markov statements of the model. For the directed path in our previous example, the conditional independence ideal of the local Markov statements is a radical ideal with 3 associated primes. However, since G is a directed tree, the conditional independence ideal of the global Markov statements is a prime ideal and it equals the vanishing ideal of G .

```

i23 : J = conditionalIndependenceIdeal (R, localMarkov G); I == J
o23 : Ideal of R
o24 = false
i25 : K = conditionalIndependenceIdeal (R, globalMarkov G); I == K
o25 : Ideal of R
o26 = true

```

The method `gaussianVanishingIdeal` computes the vanishing ideal of a Gaussian graphical model on a graph, digraph, or mixed graph. It is applied to a polynomial ring that has been created with the command `gaussianRing G` where G is either a graph, digraph, or mixed graph. In general, mixed graphs can have directed, undirected, and bidirected edges, though at present `gaussianVanishingIdeal` only works on mixed graphs that do not have undirected edges. If G is a mixed graph with no undirected edges, I_G is the vanishing ideal of the parametrization $\Sigma = (I - \Lambda)^{-T} \Psi (I - \Lambda)^{-1}$, where Σ is the variance-covariance matrix, Λ is the strictly upper triangular matrix with $\Lambda_{ij} = \lambda_{ij}$ if $i \rightarrow j$ is a directed edge in G and 0 otherwise, and Ψ is a symmetric positive definite matrix of parameters ψ_{ij} with zeros in each entry Ψ_{ij} if there is no bidirected edge in G between i and j , and $i \neq j$.

The following example computes the vanishing ideal of the Gaussian graphical model on the mixed graph with directed edges $\{1 \rightarrow 2, 1 \rightarrow 3, 2 \rightarrow 3, 3 \rightarrow 4\}$ and bidirected edges $\{1 \leftrightarrow 2, 2 \leftrightarrow 4\}$. This ideal is a principal ideal generated by one quartic polynomial with 8 terms. This ideal is not *determinantal*, i.e. it is not generated by the determinantal equations defining the trek ideal, which in this case is the zero ideal.

```

i27 : G = mixedGraph(digraph {{1,{2,3}},{2,{3}},{3,{4}}},bigraph {{1,2},{2,4}});
i28 : R = gaussianRing G; I = gaussianVanishingIdeal R;
o29 : Ideal of R
i30 : flatten degrees I
o30 = {4}
o30 : List
i31 : J = trekIdeal (R,G)
o31 = 0

```

An important problem with these models is determining which parameters are *identifiable* or *generically identifiable*, see [GPSS]. The method `identifyParameters` can be used to solve the identifiability problem for Gaussian graphical models on mixed graphs (also known as *structural equation models*). The following example shows that the parameter ψ_{24} is generically identifiable by the formula $\psi_{24} = (\sigma_{13}\sigma_{24} - \sigma_{14}\sigma_{23})/\sigma_{13}$.

```
i32 : H = identifyParameters R;
i33 : H#(p_(2,4))_0
o33 = p   s   + s   s   - s   s
      2,4 1,3   1,4 2,3   1,3 2,4
o33 : R
```

In this model there are three non-generically identified parameters. `identifyParameters` produces a hash table whose entries are indexed by the parameters and contain ideals that can be used to find explicit rational functions for every parameter that is generically identifiable.

ACKNOWLEDGMENTS. The following people generously contributed their time to the development of the package: Alexander Diaz, Shaowei Lin, David Murrugarra, and Mike Stillman. Work on the package was carried out during the 2010 and 2011 *Macaulay2* workshops, which were partially supported by the US National Science Foundation and the Institute for Mathematics and Its Applications. The authors also thank the anonymous referees for their helpful comments and suggestions that improved not only this article but also the *GraphicalModels* package.

LGP was partially supported by a 2012 SHSU Faculty Research Grant (29001). SP was partially supported by Grant #FA9550-12-1-0392 from the U.S. Air Force Office of Scientific Research (AFOSR) and the Defense Advanced Research Projects Agency (DARPA). SS was partially supported by the US National Science Foundation (DMS 0954865) and the David and Lucille Packard Foundation.

REFERENCES.

- [DSS] M. Drton, B. Sturmfels, and S. Sullivant, *Lectures on algebraic statistics*, Oberwolfach Seminars, vol. 39, Birkhäuser Verlag, Basel, 2009.
- [GPSS] L.D. García-Puente, S. Spielvogel, and S. Sullivant, *Identifying causal effects with computer algebra*, available at [arXiv:1007.3784v1\[math.ST\]](https://arxiv.org/abs/1007.3784v1).
- [GSS] L.D. Garcia, M. Stillman, and B. Sturmfels, *Algebraic geometry of Bayesian networks*, J. Symbolic Comput. **39** (2005), no. 3-4, 331–355.
- [GMS] D. Geiger, C. Meek, and B. Sturmfels, *On the toric algebra of graphical models*, Ann. Statist. **34** (2006), no. 3, 1463–1492.
- [L] S.L. Lauritzen, *Graphical models*, Oxford Statistical Science Series, vol. 17, The Clarendon Press Oxford University Press, New York, 1996.
- [M2] D.R. Grayson and M.E. Stillman, *Macaulay2, a software system for research in algebraic geometry*, available at www.math.uiuc.edu/Macaulay2/.
- [S] S. Sullivant, *Algebraic geometry of Gaussian Bayesian networks*, Adv. in Appl. Math. **40** (2008), no. 4, 482–513.
- [STD] S. Sullivant, K. Talaska, and J. Draisma, *Trek separation for Gaussian graphical models*, Ann. Statist. **38** (2010), no. 3, 1665–1685.
- [W] J. Whittaker, *Graphical models in applied multivariate statistics*, Wiley Series in Probability and Mathematical Statistics: Probability and Mathematical Statistics, John Wiley & Sons Ltd., Chichester, 1990.

RECEIVED : 2012-08-31

REVISED : 2013-01-01

ACCEPTED : 2013-03-05

lgarcia@shsu.edu : Department of Mathematics and Statistics, Sam Houston State University, Huntsville, TX 77341, USA.

Sonja.Petrovic@iit.edu : Department of Applied Mathematics, Illinois Institute of Technology, Chicago, IL 60616, USA.

smsulli2@ncsu.edu : Department of Mathematics, North Carolina State University, Raleigh, NC 27695, USA.